```
In[22]:= mean[x_] := Mean[x];

In[23]:= median[x_] := Median[x]

In[24]:= t10mean[x_] :=
          mean[Sort[x][[ Table[i, {i, Floor[0.05 Length[x]], Floor[0.95 Length[x]]}] ]] ]

In[25]:= s[x_] := Sqrt[Apply[Plus, (x - mean[x]) ^ 2] / (Length[x] - 1)]
```

sam[x, n] selects a sample of n from a list x with replacement and with equal probability.

```
In[26]:= sam[x_, n_] := x[[Table[Random[Integer, {1, Length[x]}], {i, 1, n}]] ]
```

bs[function, data] creates 2000 values of function[same size random sample from data]

```
In[27]:= bs[function_, data_] :=
           Module[{n = Length[data]}, Table[function[sam[data, n]], {i, 1, 2000}]];
```

bsCI[function, data, conf] creates a bootstrap confidence interval for function[population] for confidence level "conf" based upon 2000 same size bootstrap samples. It reports

{function name, the estimate function[data], bs confidence interval, confidence level}

```
In[28]:= bsci[function_, data_, conf_] := {function, function[data], function[data] +
             {-1, 1} Sort[ Abs[bs[function, data] - function[data] ]][[Floor[conf 2000]] ] , conf}
```

phi[r, sig] is the normal density with mean zero and s.d. "sig" evaluated at r.

```
In[29]:= phi[r_, sig_] := (1 / Sqrt[2 Pi sig ^ 2]) Exp[-r ^ 2 / (2 sig ^ 2)]
```

density[list, sig, r] is the kernel density for data "list" and kernel s.d. "sig" evaluated at r.

```
In[30]:= density[list_, sig_, r_] := mean[phi[r - list, sig]]
```

plotdensity[list, sig] plots a density for "data" and kernel s.d. "sig" over the data range.

```
In[31]:= plotdensity[list_, sig_] :=
          Module[{v = Sort[list][[{1, Length[list]}]]}, Plot[density[list, sig, r],
            {r, v[[1]], v[[2]]}, PlotRange → All, AxesOrigin → {v[[1]], 0}, PlotPoints → 2000]]
```

The few functions defined above are capable of a lot of statistical work. Some examples:

```
In[32]:= sample1 = Table[N[Sin[ Random[] ^ 3]], {i, 1, 100}];
```

The kernel density of the population estimated via the kernel density of sample1.

In[33]:=
```
plotdensity[sample1, 0.03]
```



Out[33]= - Graphics -

Above, compared to the sampling distribution of mean[sample] estimated by bootstrap.

In[34]:=
```
Show[%, plotdensity[bs[mean, sample1], 0.01]]
```



Out[34]= - Graphics -

Above compared with the actual sampling distribution of mean for samples of 100 (as estimated by drawing 1000 samples of 100 as sample1 was created).

In[35]:=
```
Show[%, plotdensity[
  Table[mean[Table[N[Sin[ Random[]^3]], {i, 1, 100}]], {i, 1, 1000}] , 0.01] ]
```





Out[35]= **-** Graphics **-**

In[36]:= **mean[sample1]**

Out[36]= 0.245872

In[37]:= **s[sample1]**

Out[37]= 0.243573

In[38]:= **%% + 1.96 {-1, 1} % / Sqrt[100]**

Out[38]= {0.198131, 0.293612}

In[39]:= **NIntegrate[Sin[r^3], {r, 0, 1}]**

Out[39]= 0.233845

In[40]:= **bsci[mean, sample1, 0.95]**

Out[40]= {mean, 0.245872, {0.198723, 0.293021}, 0.95}

In[41]:= **bsci[median, sample1, 0.95]**

Out[41]= {median, 0.212154, {0.0809763, 0.343332}, 0.95}

In[42]:= **bsci[t10mean, sample1, 0.95]**

Out[42]= {t10mean, 0.227336, {0.175867, 0.278805}, 0.95}